DAO – Data Access Object

É uma coleção de objetos que permitem manusear um Banco de Dados relacional inteiramente através de código fonte. Não há automação como quando usamos o Data Control mas em compensação o uso mais extensivo da linguagem SQL é facilitado, como por exemplo a execução de ‘Queries’ que em vez de retornar registros simplesmente cumprem uma ação (Criam um campo ou apagam um registro).

Para se usar o DAO e seus métodos e propriedades é preciso primeiramente criar variaveis que conterão o Banco de Dados, o Recordset e demais objetos a serem manipulados. Para isso deve-se dimensionar uma variável como Database. Se a opção não aparecer na lista que surge quando a palavra chave As é digitada ou um erro como ‘User defined type not defined’ ocorrer voce deve ir ao menu Project, selecionar References e marcar o item ‘Microsoft DAO 3.5 Object Library’. Note que se um Data Control já existir no projeto o passo descrito acima não é necessário.

Dim DB As Database

O escopo destas variáveis dependem da utilidade dos objetos ligados ao Banco de Dados no programa sendo desenvolvido. Como quaisquer outras elas podem valer somente para o procedimento na qual forem declaradas, para todo o formulário onde forem declaradas ou para todos os formulários do projeto se declaradas em um módulo.

Para criar um Banco de Dados vazio, sem tabelas nem campos usa-se o método CreateDatabase junto com a palavra chave Set. No exemplo abaixo VarDB é a Variável.

Set VarDB = CreateDatabase(Nome do Arq, Local, Opções)

Onde:

Nome do Arq. (String)   É o nome do Banco de Dados a ser criado, com ou sem o caminho.

Set VarDB = CreateDatabase(“C:\Meu Dir\MeuBanco.mdb”, Local, Opções)

Local (String)  É a constante dbLangGeneral para estipular que o Banco de Dados vai ser aberto usando uma lingua ocidental (Inglês, Francês, Espanhol, Português, Alemão & Italiano). Se desejar usar outra lingua no seu Banco de dados use o Help do Visual Basic para descobrir as constantes para as outras línguas. Opcionalmente pode-se concatenar uma senha a esta constante estipulando assim um controle de segurança maior ao Banco de Dados sendo criado. Lembre-se porém que se a senha for esquecida não há maneira de se abrir este Banco de Dados novamente.

Set VarDB = CreateDatabase(Nome do Arq, dbLangGeneral+”;pwd=MinhaSenha”, Opções)

Opções (Integer)  Contando que se esteja usando o VB5, é aconselhável usar a constante dbVersion30 para criar um Banco de Dados compatível com o Jet DB Engine v3.5 usado pela linguagem. As outras opções são para versões anteriores do Jet DB Engine. Há tambem uma outra constante, dbEncript, que se somada a uma das opções de versão permite criar o Banco de Dados criptografado aumentando ainda mais a segurança das informações nele contidas.

Set VarDB = CreateDatabase(Nome do Arq, Local,dbVersion30+dbEncript)

Uma vez criada a casca do Banco de dados, é preciso começar a trabalhar seu conteúdo. Para isto usaremos o Método CreateTableDef para criar a definição de uma tabela e CreateFields para criar campos para a mesma.

Deve haver uma variável dimensionada como TableDef da mesma forma como havia uma como Database. No exemplo abaixo TDEF é a variável

Dim TDEF As TableDef

A tabela também é criada usando a palavra chave Set, mais ou menos da mesma forma que o Banco de Dados. Como pode-se ver pelo exemplo abaixo CreateTableDef é um método do Database Object, por isso usamos a notação do ponto e a variável que contém o Banco de dados.

Set TDEF = VarDB.CreateTableDef(Nome)

Onde:

Nome (String)  É o nome da tabela a ser criada

Set TDEF = VarDB.CreateTableDef(“Clientes”)

Até então nós temos um Banco de Dados e uma tabela vazia, sem nenhum campo. Para criar campos usamos o método CreateField do Objeto TableDef e o Método Append também do Objeto TableDef para anexa-lo a tabela.

TDEF.Fields.Append TDEF.CreateField(Nome Campo, Tipo Dados, Tamanho)

Pode-se também definir um campo como uma variável para que possa ser anexado a mais de uma tabela no mesmo ou em diferentes Bancos de Dados. 

Dim Campo As Field

.

.

Set Campo = TDEF.CreateField(Nome Campo, Tipo Dados, Tamanho)

TDEF.Fields.Append Campo

Onde:

Nome do Campo (String)  É o nome do campo a ser criado.

TDEF.Fields.Append TDEF.CreateField(“Endereço”, Tipo Dados, Tamanho)

Tipo de Dados (Inteiro Longo)  Uma dentre as constantes na tabela abaixo para definir o tipo de dados que o campo a ser criado vai conter.

Constante
Tipo
Constante
Tipo
Constante
Tipo

dbBigInt
Big Integer
dbDecimal
Decimal
dbMemo
Memo

dbBinary
Binary
dbDouble
Double
dbNumeric
Numeric

dbBoolean
Boolean
dbFloat
Float
dbSingle
Single

dbByte
Byte
dbGUID
GUID
dbText
Text

dbChar
Char
dbInteger
Integer
dbTime
Time

dbCurrency
Currency
dbLong
Long
dbTimeStamp
Time Stamp

dbDate
Date/Time
dbLongBinary
OLE Object
dbVarBinary
VarBinary

TDEF.Fields.Append TDEF.CreateField(Nome do Campo, dbText, Tamanho)

Tamanho (Integer)  É o tamanho do campo se ele for texto (dbText).

TDEF.Fields.Append TDEF.CreateField(Nome do Campo, dbText, 50)

Se for necessário alterar propriedades de um Objeto Fields, acesse-o pelo seu ordinal. Este é definido pela ordem de execução, ou seja, a primeira linha é executada primeiro então este passa a ser o primeiro campo ou Campo 0.

TDEF.Fields(0).Attributes = dbAutoIncrField

O exemplo acima transforma o primeiro campo de uma tabela num campo de auto incremento.

Para concluir a construção da tabela é necessário anexá-la ao Banco de Dados por intermédio do Método Append do Objeto TableDefs.

DB.TableDefs.Append TDEF

Para se criar uma chave primária pode-se usar tanto um método do DAO como um comando em SQL. Sendo que o SQL é mais fácil de usar e resolve o problema sem a criação de mais variáveis e objetos. O SQL neste caso é chamado de ‘Action Query’ porque a expressão não retornará nenhum registro ou Recordset mas executará um comando que alterará a estrutura da Tabela.

O SQL será enviado através do Método Execute do DAO com a seguinte sintaxe:

DB.Execute Ação, Opções

Onde:

Ação (String)  É o SQL a ser executado. Veja a Apostila de SQL para detalhes de Sintaxe e comandos

Opções (Long)  Opcional, determina o tipo de integridade a ser imposta. Use uma ou mais das  seguintes constantes. Se usar mais de uma some-as.

Constante
Descrição

dbDenyWrite
Não permite que outros usuarios gravem no Banco de Dados

dbInconsistent
{Default} Atualização Inconsistente (*) 

dbConsistent
Atualização Consistente (**) 

dbSQLPassThrough
Passa a pesquisa para ser processada por um Banco de Dados ODBC

dbFailOnError
Cancela atualizações já feitas na mesma pesquisa se ocorrer um erro

DbSeeChanges
Gera erro se outro usuário estiver mudando os mesmos dados que voce

(*) Inconsistente  Estado de um Recordset de tabelas múltiplas que permite a atualização de todos os campos, mesmo que num Recordset criado de um Join de 1 – M  voce queira gravar um valor na chave estrangeira que não exista na primária. Se integridade referencial estiver imposta a atualização falhará.

(**) Consistente  O estado de um Recordset onde uma visão consistente dos dados é imposta. A operação descrita acima falharia de qualquer maneira.

O programa abaixo cria um Banco de Dados chamado Teste.MDB com uma tabela chamada Clientes que possui os seguintes campos :

ID 

 Inteiro Longo (Auto Numeração) Chave Primária

Nome 

 Texto – 30 Caracteres

Endereço 
 Texto – 50 Caracteres

Tel 

 Texto – 15 Caracteres

Cidade 
 Texto – 25 Caracteres

Idade 

 Byte


Dim DB As Database      '' Variável para conter o Banco de Dados

Dim TDEF As TableDef    '' Variável para conter a definição da tabela


Private Sub Form_Load()

'' Cria Banco de Dados com senha

Set DB = CreateDatabase("C:\Teste.mdb", dbLangGeneral + ";pwd=Senha")

'' Cria Tabela Clientes

Set TDEF = DB.CreateTableDef("Clientes")

'' Cria campos para a tabela Clientes

TDEF.Fields.Append TDEF.CreateField("ID", dbLong)

'' Primeiro campo será chave primária

'' Portanto deve ser Inteiro longo e ter o atributo

'' Auto Numeração

TDEF.Fields(0).Attributes = dbAutoIncrField

TDEF.Fields.Append TDEF.CreateField("Nome", dbText, 30)

TDEF.Fields.Append TDEF.CreateField("Endereço", dbText, 50)

TDEF.Fields.Append TDEF.CreateField("Tel", dbText, 15)

'' Para suprimir o erro 'Invalid Use of Null' neste campo

TDEF.Fields(3).AllowZeroLength = True

TDEF.Fields(3).Required = False

'' Para suprimir o erro 'Invalid Use of Null' neste campo

TDEF.Fields.Append TDEF.CreateField("Cidade", dbText, 25)

TDEF.Fields(4).AllowZeroLength = True

TDEF.Fields(4).Required = False

'' Para suprimir o erro 'Invalid Use of Null' neste campo

TDEF.Fields.Append TDEF.CreateField("Idade", dbByte)

TDEF.Fields(5).AllowZeroLength = True

TDEF.Fields(5).Required = False

'' Concretiza a criação da Tabela e seus campos

DB.TableDefs.Append TDEF

'' Cria indice no campo ID, transformando-o em Chave primária

DB.Execute "create index IND on clientes (id) with primary"

'' Fecha o Banco de Dados

DB.Close

End Sub


As propriedades AllowZeroLenght e Required do Objeto Fields são usadas nos campos Tel, Cidade e Idade para permitir tanto comprimento zero (“”) como a ausência de dados válidos (Null).

AllowZeroLength (Boolean)  Se Verdadeiro (True) o campo pode conter uma string vazia. Somente quando o campo for String.

Required (Boolean)  Se verdadeiro (True) o campo não pode conter um Null, ou seja, nenhum dado válido, nem mesmo uma string vazia ou um zero.


Para se acessar um Banco de Dados existente usando Recordsets como no Data Control, deve-se criar variáveis que representaram estes objetos. Os métodos e propriedades do DAO e do Data Control para criar e acessar Recordsets são bastante similares.

Dim VarDB as Database

Dim VarRS as Recordset

Para abrir um Banco de Dados usa-se o método OpenDatabase com o comando Set na variável que foi declarada para conter o Banco de Dados com a seguinte sintaxe:

Set VarDB=OpenDatabase(Nome do Arq, Opções, Leitura, Conexão)

Onde:

Nome do Arq (string)  É o nome do Banco de Dados com ou sem o caminho.

Set DB = OpenDatabase("c:\teste.mdb", Opções, Leitura, Conexão)

Opções (Boolean)  Opcional. Se verdadeiro (True) Banco de Dados é aberto para um só usuario (modo Exclusivo) (*).

Set VarDB=OpenDatabase(Nome do Arq, False, Leitura, Conexão)

Leitura (Boolean)  Opcional. Se verdadeiro (True) o Banco de Dados é aberto somente para leitura, atualizações não poderão ser feitas.

Set VarDB=OpenDatabase(Nome do Arq, Opções, False, Conexão)

Conexão (String)  Opcional. Corresponde a propriedade Connection. Entre seus usos esta o de entrar a senha do Banco de Dados.  

Set VarDB=OpenDatabase(Nome do Arq, Opções, Leitura, ";pwd=Senha")

(*) Serve somente para arquivos do Microsoft Jet Database Engine

Para se abrir um Recordset o processo é bem similar ao descrito acima, mudando apenas o comando e alguns de seus parâmetros. Em vez de OpenDatabase usa-se OpenRecordset na variável que foi declarada para conter o Recordset  com a seguinte sintaxe:

Set VarRS  = OpenRecordset(Fonte, Tipo, Opções, Tranca)

Onde:

Fonte (string)  Nome de uma tabela no Banco de Dados aberto ou comando SQL.

Set VarRS  = OpenRecordset(“Clientes”, Tipo, Opções, Tranca)

Tipo (Integer)  Opcional. Determina o tipo de Recordset a ser aberto usando uma das constantes da tabela abaixo.

Constante
Descrição

dbOpenTable
Abre um Recordset tipo Table (Somente Jet Engine)

DbOpenDynaset
(Default) Abre um Recordset tipo Dynaset

dbSnapshot
Abre um Recordset tipo Snapshot

dbOpenForwardOnly
Abre um Recordset tipo Forward Only

Set VarRS  = OpenRecordset(“Clientes”, DbOpenDynaset, Opções, Tranca)

Opções (Long)  Opcional. Combinação de constantes que determina caracteristicas do Recordset a ser aberto. Vide tabela

Constante
Descrição

dbDenyWrite
Não permite que outros usuarios gravem no Banco de Dados

dbInconsistent
{Default} Atualização Inconsistente (*) 

dbConsistent
Atualização Consistente (**) 

dbSQLPassThrough
Passa a pesquisa para ser processada por um Banco de Dados ODBC

dbFailOnError
Cancela atualizações já feitas na mesma pesquisa se ocorrer um erro

dbSeeChanges
Gera erro se outro usuário estiver mudando os mesmos dados que voce

dbAppendOnly
Permite a adição de novos registros mas não a modificação ou exclusão

dbDenyRead
Não permite que outros usuários leiam o Banco de Dados 

(*) Inconsistente  Estado de um Recordset de tabelas múltiplas que permite a atualização de todos os campos, mesmo que num Recordset criado de um Join de 1 – M  voce queira gravar um valor na chave estrangeira que não exista na primária. Se integridade referencial estiver imposta a atualização falhará.

(**) Consistente  O estado de um Recordset onde uma visão consistente dos dados é imposta. A operação descrita acima falharia de qualquer maneira.

Set VarRS  = OpenRecordset (Fonte, Tipo, dbSeeChanges, Tranca)

Tranca (Integer)  Opcional. Determina o Trancamento do Recordset de acordo com a tabela abaixo.

Constante
Descrição

dbReadOnly
Não deixa mudanças serem efetuadas

dbPessimistic
(Default) Tranca o registro assim que voce começa a editar

dbOptimistic
Só tranca o registro quando houver o update

O programa abaixo lê o primeiro registro da tabela Clientes do Banco de Dados criado no programa exemplo anterior e mostra os campos em 5 Labels se algum registro tiver sido acrescentado. Supõe-se que hajam 5 Labels com seus nomes padrão no formulario do projeto (label1, Label2...)

Dim DB As Database      '' Variável para conter o Banco de Dados

Dim RS As Recordset     '' Variável para conter o Recordset


Private Sub Form_Load()

'' Abre Banco de Dados

Set DB = OpenDatabase("c:\teste.mdb", False, False, ";pwd=Senha")

'' Abre Recordset

Set RS = DB.OpenRecordset("Clientes", dbOpenDynaset)

'' Verifica se há registros para então mostrar o primeiro

If RS.RecordCount Then

                      RS.MoveFirst

                      Label1 = RS.Fields(1)

                      Label2 = RS.Fields(2)

                      Label3 = RS.Fields(3)

                      Label4 = RS.Fields(4)

                      Label5 = RS.Fields(5)

                End If

End Sub

